

# 応用数値解析特論 第7回

～プログラミング言語 FreeFem++～

かつらだ まさし  
桂田 祐史

<https://m-katsurada.sakura.ne.jp/ana2022/>

2022年11月14日

# 目次

- ① 本日の内容
- ② FreeFem++の文法
  - はじめに
  - 汎用のプログラミング機能
    - C言語と良く似ているところ
    - データ型
    - 配列型
  - 有限要素法のための機能
    - 有限要素法のプログラムの構成
    - 領域の定義と領域の三角形分割
    - 有限要素空間
    - 弱形式を定義して解く
- ③ 参考プログラム — 有限要素解の誤差を見る
  - 図4の作成方法
- ④ 参考文献

# 本日の内容

- GLSC3D が使えるように、少し検討タイム  
授業 WWW サイト

<https://m-katsurada.sakura.ne.jp/ana2022/>

の連絡事項にチェック用のコマンド

- FreeFem++ の文法の説明

# 7 FreeFem++の文法

## 7.1 はじめに

FreeFem++ は有限要素法によって微分方程式の数値シミュレーションを行うためのソフトウェアであり、言語処理系である (Hecht [1])。

インタプリタである (その点は MATLAB や Python と似ている)。

今回は、プログラミング言語としての FreeFem++ を説明する (マニュアルを見ても良く分からない — 少なくとも私は)。

FreeFem++ のことを「有限要素法専用ツール」と考える人もいる。確かに有限要素法に便利な命令が組み込まれているが、それ以外の目的のプログラミングに必要な機能も十分に備わっている (実際、有限体積法や差分法のプログラムも記述可能である)。効率を度外視すれば、C のようなプログラミング言語で出来ることは FreeFem++ でも出来る、と考えよう。

文法は、C++ に似ている (ゆえに C にも似ている)。C しか知らない人は、**C++ のストリームを使った入出力** (cout, cin の利用) を調べておくこと ([2] の付録に書いておいた)。

参考: FreeFem++ は C++ で記述されている。

マニュアル Hecht[3] は事例集の性格が強く、言語仕様が整理された形では載っていない。以下の説明は、個人的なノートである桂田 [2] に基づく。

マニュアル以外の情報源として、日本応用数理学会のチュートリアル (鈴木 [4], [5])、テキスト大塚・高石 [6] などがある。

## 7.1 はじめに 基本的な Poisson 方程式のプログラム

<https://m-katsurada.sakura.ne.jp/program/fem/poisson.edp> に置いてある。

```
// poisson.edp
// 境界の定義 (単位円), いわゆる正の向き
border Gamma(t=0,2*pi) { x=cos(t); y=sin(t); }
// 三角形要素分割を生成 (境界を 50 に分割)
mesh Th = buildmesh(Gamma(50));
plot(Th,wait=true); // plot(Th,wait=true,ps="Th.eps");
// 有限要素空間は P1 (区分的 1 次多項式) 要素
real [int] levels =-0.012:0.001:0.012;
fespace Vh(Th,P1);
Vh u,v;
// Poisson 方程式  $-\Delta u=f$  の右辺
func f = x*y;
// 問題を解く
solve Poisson(u,v)
  = int2d(Th) (dx(u)*dx(v)+dy(u)*dy(v))-int2d(Th) (f*v)
  +on(Gamma,u=0);
// 可視化 (等高線)
plot(u,wait=true);
//plot(u,viso=levels,fill=true,wait=true);
// 可視化 (3 次元) --- マウスで使って動かせる
plot(u,dim=3,viso=levels,fill=true,wait=true);
```

→ 独特の命令ばかりで、汎用のプログラミング言語の機能があることは分かりにくい。

## 7.2 汎用のプログラミング機能

### 7.2.1 C 言語と良く似ているところ

改めて数えるととても多い。

- // から行末までは注釈、/\* と \*/ で挟まれた部分は注釈 (共に C 言語と同じ)
- 文の最後は ;
- 四則演算 (+, -, \*, /) や、代入 (=) などの演算子
- 0 は偽、0 以外の整数は真とみなす。一方、比較演算・論理演算などの結果は 0 (false) または 1 (true).
- 変数宣言の文法も C 言語と同様。型名の後に, で区切った名前のリストを書く。
- 関数呼び出しの文法も C 言語と同様。
- ブロックは { と } で複数 (0 個以上) の文を囲んで作る。
- 比較演算子 (==, !=, <, <=, >, >=)、論理演算子 (&&, ||, !)、if, if else などの制御構造。  
ただし switch はない。
- for, while などの繰り返し制御。break (ループを抜ける), continue (次の繰り返し) など。  
ただし do while はない。
- 数学関数の名前

他にもあるだろう…

## 7.2.2 データ型

- 整数を表すための `int` がある (C 言語の `int` に相当)
- 実数を表すための `real` がある (C 言語の `double` に相当)
- 複素数を表すための `complex` がある (C 言語の `complex` に相当, 実部・虚部が `double`)  
例えば `complex a=1+2i;` 入出力は 2次元ベクトル風の (1,2)
- 論理を表すための `bool` がある (C 言語の `bool` に相当). `true`, `false` という値があるが、それぞれ 1, 0 の別名と考えて良い。  
例えば `plot(u,wait=true);` は `plot(u,wait=1);` と同じ。
- 文字列を表すための `string` がある (C++言語の `string` に相当, 日本語不可?)。
  - 2つの `string` `s1`, `s2` を、(+ 演算子を用いて) `s1+s2` で連結できる。
  - `string+数値` とすると、数値を文字列に変換してから連結する。

```
real a=1.23, b=4.56;  
string s;  
s= "a=" + a + ", b=" + b + ".";  
cout << s << endl;
```

- `string` を `int` に変換する `atoi()`, `string` を `real` に変換する `atof()` がある (C 言語の真似)。

## 7.2.3 配列型 1次元

1次元配列は、C言語に(少し)似ている。

配列  $a$  の第  $i$  要素は、 $a[i]$  としてアクセスできるが、 $a(i)$  としてアクセスするのが普通？

```
real[int] a1(3); // Cで double a1[3]; とするの似ている
for (int i=0;i<3;i++)
    a1(i)=i; // a1[i]=i; としても良い。
```

```
real[int] a2 = [0,1,2]; // Cで double a[]={0,1,2}; とするの似てる
real[int] a3 = 0:2; // これは少し MATLAB 風
```

```
cout << "a1=" << a1 << endl;
cout << "a2=" << a2 << endl;
cout << "a3=" << a3 << endl;
```

追記:  $a1$  の要素数は  $a1.n$  で得られる。



## 7.2.3 配列型 2次元

2次元配列は少し違う。2次元配列  $a$  の  $(i,j)$  要素にアクセスするには  $a(i,j)$  とする。

```
real[int,int] kuku(9,9);
int i,j;
for (i=0; i<kuku.n; i++) {
    for (j=0; j<kuku.m; j++) {
        kuku(i,j)=(i+1)*(j+1);
        cout << setw(3) << kuku(i,j);
    }
    cout << endl;
}
cout << kuku << endl;

real[int,int] kuku2=[[1,2,3,4,5,6,7,8,9],
                    [2,4,6,8,10,12,14,16,18],
                    [3,6,9,12,15,18,21,24,27],
                    [4,8,12,16,20,24,28,32,36],
                    [5,10,15,20,25,30,35,40,45],
                    [6,12,18,24,30,36,42,48,54],
                    [7,14,21,28,35,42,49,56,63],
                    [8,16,24,32,40,48,56,64,72],
                    [9,18,27,36,45,54,63,72,81]];

cout << kuku2 << endl;
```

## 7.3 有限要素法のための機能

### 7.3.1 有限要素法のプログラムの構成

有限要素法のプログラムと言っても色々あるが、基本的と考えられる2次元 Poisson 方程式の境界値問題のプログラムを例にして説明する。

- ① 領域の定義と領域の三角形分割
- ② 有限要素空間の定義
- ③ 弱形式を次のいずれかで定義して解く。
  - a `solve()`  
弱形式を与えると同時にそれを解く (弱解を求める)。
  - b `problem()`  
弱形式を与えて問題を解く関数を定義する。発展問題で便利。
  - c `varf, matrix`  
弱形式を与えて連立1次方程式を作る。

## 7.3.2 領域の定義と領域の三角形分割 (2次元の場合)

問題を考える領域を定義し、三角形分割をすることが必要である。

- 2次元 (有界) 領域の多くは、その境界曲線を定義することで定まる。
- 境界曲線は `border` という型の変数として定義される。
- 三角形分割は `mesh` という型の変数として定義される。
- `buildmesh()` という関数は、各 `border` を何等分するか指定することで、`border` の囲む領域を三角形分割して、`mesh` 型のデータを作る。
- `mesh` 型のデータは、`readmesh()`, `writemesh()` という関数を用いて入出力できる (フォーマットはテキスト・ファイル)。
- `mesh` 型のデータは、`plot()` により可視化できる。
- 矩形領域 (辺が座標軸に平行な長方形) は、`square()` という命令で `mesh` 型データが作れる (参考「[FreeFem++ノート](#)」)。

円周全体を  $C$  とする

```
border C(t=0,2*pi) { x=cos(t); y=sin(t); }
```

円周の上半分、下半分を別々に  $\Gamma_1, \Gamma_2$  と定義する

```
int C=1;
...
border Gamma1(t=0,pi) { x=cos(t); y=sin(t); label=C; }
border Gamma2(t=pi,2*pi) { x=cos(t); y=sin(t); label=C; }
```

正方形領域  $(0,1) \times (0,1)$  の4つの辺  $C_1, C_2, C_3, C_4$  を定義

```
border C1(t=0,1) { x=t; y=0; label=1; }
border C2(t=0,1) { x=1; y=t; label=2; }
border C3(t=0,1) { x=1-t; y=1; label=3; }
border C4(t=0,1) { x=0; y=1-t; label=4; }
```

(label の値指定は必須ではない。指定する場合は 0 以外の値を選ぶ。)

それぞれ表示してみる

// 例 1




```
border C(t=0,2*pi) { x=cos(t); y=sin(t); }
mesh Th1=buildmesh(C(50));
plot(Th1,wait=true,ps="Th1.eps");
```

// 例 2

```
int C0=1;
border Gamma1(t=0,pi) { x=cos(t); y=sin(t); label=C0; }
border Gamma2(t=pi,2*pi) { x=cos(t); y=sin(t); label=C0; }
mesh Th2=buildmesh(Gamma1(25)+Gamma2(50));
plot(Th2,wait=true,ps="Th2.eps");
```

// 例 3

```
border C1(t=0,1) { x=t; y=0; label=1; }
border C2(t=0,1) { x=1; y=t; label=2; }
border C3(t=0,1) { x=1-t; y=1; label=3; }
border C4(t=0,1) { x=0; y=1-t; label=4; }
mesh Th3=buildmesh(C1(20)+C2(20)+C3(20)+C4(20));
plot(Th3,wait=true,ps="Th3.eps");
```





図 1:  $C(50)$ 図 2:  $\text{Gamma1}(25)+\text{Gamma2}(50)$ 図 3:  $C1(20)+C2(20)+\dots$

## 7.3.2 領域の定義と領域の三角形分割 メッシュ(mesh)

有限個の Jordan 閉曲線で囲まれた多重連結領域を三角形分割することもできる。

sampleMesh.edp

```
border a(t=0,2*pi){ x=cos(t); y=sin(t);label=1;}  
border b(t=0,2*pi){ x=0.3+0.3*cos(t); y=0.3*sin(t);label=2;}  
plot(a(50)+b(+30),wait=true,ps="border.eps");  
mesh ThWithoutHole = buildmesh(a(50)+b(+30));  
plot(ThWithoutHole,wait=1,ps="Thwithouthole.eps");  
plot(a(50)+b(-30),wait=true,ps="borderminus.eps");  
mesh ThWithHole = buildmesh(a(50)+b(-30));  
plot(ThWithHole,wait=1,ps="Thwithhole.eps");
```



## 7.3.2 領域の定義と領域の三角形分割    メッシュ(mesh)

普通は mesh データの細かいことは見る必要がないかもしれないが…

- `Th` をメッシュとするとき、`Th.nt` は三角形の数 (the number of triangles)、`Th.nv` は節点の数 (the number of vertices)、`Th.area` は領域の面積 (area) である。
- `Th(i)` は  $i$  番目の節点 ( $i = 0, 1, \dots, \text{Th.nv} - 1$ ) で、その座標は `Th(i).x` と `Th(i).y` である。`Th(i).label` はその節点のラベル (領域内部にあれば 0, それ以外は境界のどこか) を表す。
- `Th[i]` は  $i$  番目の三角形 ( $i = 0, 1, \dots, \text{Th.nt} - 1$ )、`Th[i][j]` は  $i$  番目の三角形の  $j$  番目の節点 ( $j = 0, 1, 2$ ) の全体節点番号、その節点の座標は `Th[i][j].x` と `Th[i][j].y` である。三角形の面積は `Th[i].area` である。
- 点  $(x, y)$  を含む三角形の番号は `Th(x, y).nuTriangle` で得られる。



## 7.3.2 領域の定義と領域の三角形分割      メッシュ(mesh)

次のような場合に `readmesh()`, `writemesh()` は有効である。

- ① FreeFem++ を用いて三角形分割を行い、得られたメッシュ・データを外部のプログラムで利用する (有限要素解の計算は自作プログラムで行う等)。
- ② 自作のプログラムで三角形分割を行い、そのメッシュ・データを FreeFem++ で利用する。

`readmesh()`, `writemesh()` で入出力される **メッシュ・データのフォーマット** については、「FreeFem++ノート §6.2 mesh ファイルの構造」を見よ。

- Mesh 型の変数 `Th` の内容は、

```
writemesh(Th, "bunkatsu.msh");
```

のようにしてファイルに出力できる。
- そのフォーマットにのっとして作られたファイルがあれば、

```
Mesh Th=readmesh("bunkatsu.msh");
```

のようにして読み込める。

## 7.3.3 有限要素空間

既に定義しておいた mesh 型データと、要素の種類を表す名前 (P1, P2, ...) を用いて、有限要素空間 (この講義では  $\tilde{X}$  のように表したが、 $V_h$  などの記号で表すことが多い) を定義する。

fespace 型の変数は関数空間を表すことになる。

例えば Th という mesh 型の変数があるとき、

```
fespace Vh(Th,P1);
```

とすると有限要素空間  $V_h$  が定義される。

これは文法的には型名で

```
Vh u,v;
```

として変数  $u, v$  が定義できる。これらが個々の関数を表す。

(数学語では  $u, v \in V_h$  という調子)

(注 これまでの授業で、三角形要素分割して、区分的 1 次多項式 (P1 要素) しか紹介しなかったが、Poisson 方程式の境界値問題以外では、他の要素 (P1b, P2, P2Morley,...) が必要になることがある。)

## 7.3.3 有限要素空間

- $u$  の節点での値を集めた配列は  $u[]$  で表す。  
 $u[].n$  ( $u.n$  でも同じ) は  $Th.nv$  と同じである。  
 $i$  番目の節点での値 (授業中の式で  $u^i = \hat{u}(P_i)$ ) は  $u[](i)$
- $u$  は補間多項式でもあり、 $(x, y)$  での値は  $u(x, y)$  で得られる。

## 7.3.3 弱形式を定義して解く

いよいよ弱形式を定義する方法の説明である。大きく分けて3通りある。

- Ⓐ solve — 弱形式を与えると同時にそれを解く (弱解を求める)。
- Ⓑ problem — 弱形式を与えて問題を解く関数を定義する。
- Ⓒ varf, matrix — 弱形式を与えて連立1次方程式を作る。

これまで説明して来た次の Poisson 方程式の境界値問題を元に説明する。

$$(1a) \quad -\Delta u(x, y) = f(x, y) \quad ((x, y) \in \Omega)$$

$$(1b) \quad u(x, y) = g_1(x, y) \quad ((x, y) \in \Gamma_1)$$

$$(1c) \quad \frac{\partial u}{\partial \mathbf{n}}(x, y) = g_2(x, y) \quad ((x, y) \in \Gamma_2).$$

弱解  $u$  とは、 $X_{g_1}$  に属し、次の弱形式を満たすものである。

$$(2) \quad \langle u, v \rangle = (f, v) + [g_2, v] \quad (v \in X).$$

ただし

$$(3) \quad X_{g_1} := \{w \mid w = g_1 \text{ on } \Gamma_1\}, \quad X := \{v \mid v = 0 \text{ on } \Gamma_1\}.$$

## 7.3.3 弱形式を定義して解く (a) solve を利用

Poisson 方程式の境界値問題を解くサンプル・プログラム `poisson.edp` では、(a) を用いた。

— solve で弱形式を定義して解く —

```
solve Poisson(u,v)=
  int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))-int2d(Th)(f*v)-int1d(Th,2,3)(g2*v)
  +on(1,4,u=g1);
```

次はどの方法でも共通である。

- `dx()`, `dy()` はそれぞれ  $x$ ,  $y$  での微分を表す。  
高階の微分は `dxx()`, `dxy()`, `dyy()` のようにする。
- `int2d(Th)` は、`Th` の領域全体の積分 (重積分) を表す。  
また `int1d(Th,2,3)` は境界のうち、ラベルが 2,3 である部分 (正方形の右と上) の積分 (境界積分、今の場合は線積分) を表す。
- `+on(1,4,u=g1)` は境界のうち、ラベルが 1,4 である部分 (正方形の下と左) で、 $u = g_1$  という Dirichlet 境界条件を課すことを表す (`+on(1,u=g1)+on(4,u=g1)` と分けて書くことも可能)。  
ベクトル値関数の場合は、`+on(1,u1=g1,u2=g2)` のように複数の方程式を書くこともできる。

以下、この問題の場合に、(b), (c) がどうなるか示す。

## 7.3.3 弱形式を定義して解く (b) problem を利用

(b) problem を利用する方法では、次のようになる。

problem で弱形式を定義して解く

```
problem Poisson(u,v)=  
  int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))-int2d(Th)(f*v)-int1d(Th,2,3)(g2*v)  
  +on(1,4,u=g1);  
  
Poisson;
```

この問題の場合は、`solve` と比べての利点は特に感じられないかもしれないが、時間発展の問題では、同じ形の弱形式を何度も解く必要が生じるので、有効である (効率が上がる可能性がある — 後述)。

## 7.3.3 弱形式を定義して解く (c) varf, matrix を利用

varf, matrix を利用

```
real Tgv=1.0e+30; // tgv と小文字でも可
varf a(u,v)=
  int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))
  +on(1,4,u=g1);
matrix A=a(Vh,Vh,tgv=Tgv,solver=CG);
varf l(UNUSED,v)=
  int2d(Th)(f*v)+int1d(Th,2,3)(g2*v)
  +on(1,4,UNUSED=0);
Vh F;
F[]=1(0,Vh,tgv=Tgv);
u[]=A^-1*F[];
```

あらすじは、連立1次方程式  $A\mathbf{u} = \mathbf{f}$  の  $A$ ,  $\mathbf{f}$  を別々に計算して、 $A^{-1}\mathbf{f}$  を計算することで  $\mathbf{u}$  を得る、ということである (詳細は、実は現時点で把握していないので省略する)。

tgv (terrible great value) は以前説明した ( $tgv = 10^{30}$ )。

solver= は連立1次方程式の解法を指定する。

CG

**CG 法** (共役勾配法) (反復法, 正値対称行列 (*spd*) 用)

GMRES

**GMRES 法** (反復法, 一般の正則行列用)

UMFPACK

**UMFPACK** を利用 (直接法, 一般の正則行列用)

sparsesolver

ダイナミック・リンクで**外部のソルバー**を呼ぶ

## 参考プログラム — 有限要素解の誤差を見る

有限要素解の収束、誤差の減衰は本科目の最後に説明する予定であるが、見ておこう。  
真の解  $u$ , 有限要素解  $\hat{u}$  について

$$\|u - \hat{u}\|_{L^2} = \left( \iint_{\Omega} |u(x, y) - \hat{u}(x, y)|^2 dx dy \right)^{1/2}$$

を  $L^2$  誤差、

$$\|u - \hat{u}\|_{H^1} = \left( \|u - \hat{u}\|_{L^2}^2 + \|u_x - \hat{u}_x\|_{L^2}^2 + \|u_y - \hat{u}_y\|_{L^2}^2 \right)^{1/2}$$

を  $H^1$  誤差と呼ぶ。

領域の分割を細かくしたとき、これらの誤差がどのように減衰するか、厳密解が分かる問題で調べてみよう。

```
curl -O https://m-katsurada.sakura.ne.jp/program/fem/poisson-mixedBC-mk.edp
FreeFem++ poisson-mixedBC-mk.edp
FreeFem++ poisson-mixedBC-mk.edp 2
FreeFem++ poisson-mixedBC-mk.edp 4
...
FreeFem++ poisson-mixedBC-mk.edp 64
```

正方形の辺を  $20 \cdot 2^m$  ( $m = 0, 1, \dots, 6$ ) 分割したときの誤差を近似計算する。



## 参考プログラム — 有限要素解の誤差を見る

(授業後の書き足し)

このプログラムで解いている問題は、先に厳密解 (真の解)  $u$  を選んで、それから  $f$ ,  $g$ ,  $h$  を計算して作ったのであろう。

$\Omega = (0, 1) \times (0, 1)$ ,  $\Gamma_1 = \{(1, y) \mid 0 \leq y \leq 1\} \cup \{(x, 1) \mid 0 \leq x \leq 1\} \cup \{(0, y) \mid 0 \leq y \leq 1\}$ ,  
 $\Gamma_2 = \{(x, 0) \mid 0 \leq x \leq 1\}$ .

$$u(x, y) = \sin(\pi x) \sin \frac{\pi y}{2}$$

より

$$f(x, y) = -\Delta u(x, y) = \frac{5\pi^2}{4} \sin(\pi x) \sin \frac{\pi y}{2},$$

$$g(x, y) = u(x, y) = \sin(\pi x) \sin \frac{\pi y}{2},$$

$$h(x, y) = \frac{\partial u}{\partial n}(x, 0) = -\frac{\partial u}{\partial y}(x, 0) = -\frac{\pi}{2} \sin(\pi x) \cos \frac{\pi y}{2} \Big|_{y=0} = -\frac{\pi}{2} \sin(\pi x).$$

# 参考プログラム — 有限要素解の誤差を見る




図 4: 1 辺を  $n = 20, 40, 80, \dots, 1280$  分割したときの誤差 (横軸  $n$ )

$L^2$  誤差は  $O(n^{-2})$ ,  $H^1$  誤差は  $O(n^{-1})$  となっている (実は理論からの予想と一致)。

## 図4の作成方法

データは数が少ないので、一つ一つの分割数について、`poisson-mixedBC.edp` を使って誤差を得た。それを次のように記録したファイルを作った。(工夫すれば自動化できるであろう。)

error.txt

```
20 0.00162987 0.102169
40 0.000408387 0.0511309
80 0.000102155 0.0255713
160 2.55422e-05 0.0127864
320 6.38579e-06 0.00639328
640 1.59646e-06 0.00319665
1280 3.99119e-07 0.00159833
```

error.gp … gnuplot でグラフを描き、イメージ・データも作る

```
set logscale
set format y "10^{%L}"
set format x "10^{%L}"
plot [10:2000] "error.txt" using 1:2 with lp title "L2 error", \
    "error.txt" using 1:3 with lp title "H1 error"
set term png
set output "error.png"
replot
```

1, 4-5 行目が必須。2,3 行目は凡例の体裁を整える。最後の3行はイメージ・データ作成のため。ターミナルで `gnuplot error.gp` とすれば、グラフが描かれ、`error.png` が出来る。

## 図4 の作成方法

ちなみに (このやり方を勧めているわけではないが)、次のようなシェル・スクリプトを実行して `error.txt` を作成した (データ作成の自動化)。

```
make-data.sh
```

```
#!/bin/sh
rm -f error2.txt
for i in 1 2 4 8 16 32 64
do
    echo ${i}
    FreeFem++ poisson-mixedBC.edp ${i} | grep '^n=' | grep H1 >> error2.txt
done
sed 's/n=//;s/L2-error=//;s/H1-error=/' error2.txt>error.txt
```

```
chmod +x make-data.sh
./make-data.sh
```

データ作成を自動化すると、作成手順の記録が残り、結果の再現も容易になる。記録を残し、再現性を確保することは非常に重要であり、工夫すべきである。

関数定義の話。  
gnuplot の紹介

- [1] Hecht, F.: New development in FreeFem++, *J. Numer. Math.*, Vol. 20, No. 3-4, pp. 251–265 (2012), <https://www.um.es/freefem/ff++/uploads/Main/NewDevelopmentsInFreeFem.pdf>.
- [2] 桂田祐史：FreeFEM++ ノート,  
<https://m-katsurada.sakura.ne.jp/labo/text/freefem-note.pdf>  
(2012～).
- [3] Hecht, F.: Freefem++,  
<https://doc.freefem.org/pdf/FreeFEM-documentation.pdf>, 以前は  
<http://www3.freefem.org/ff++/ftp/freefem++doc.pdf> にあった。
- [4] Suzuki, A.: Finite element programming by FreeFem++ —intermediate course, 日本応用数学会「産業における応用数理」研究部会のソフトウェアセミナー「FreeFem++ による有限要素プログラミング — 中級編 —」(2016/2/11-12) の配布資料で、<https://www.ljll.math.upmc.fr/~suzukia/FreeFempp-tutorial-JSIAM2016/> から入手できる (2016).

- [5] Suzuki, A.: Finite element programming by FreeFem++ —advanced course, 日本応用数理学会「産業における応用数理」研究部会のソフトウェアセミナー「FreeFem++ による有限要素プログラミング — 中級編 —」(2016/6/4-5)の配布資料で、<https://www.ljll.math.upmc.fr/~suzukia/FreeFempp-tutorial-JSIAM2016b/> から入手できる (2016).
- [6] 大塚厚二, 高石武史: 有限要素法で学ぶ現象と数理 — FreeFem++ 数理思考プログラミング —, 共立出版 (2014), <https://sites.google.com/a/comfos.org/comfos/ffempp> というサポート WWW サイトがある. Maruzen eBook に入っているので、<https://elib.maruzen.co.jp/elib/html/BookDetail/Id/3000018545> でアクセス出来る.